

G01EAF – NAG Fortran Library Routine Document

Note. Before using this routine, please read the Users' Note for your implementation to check the interpretation of bold italicised terms and other implementation-dependent details.

1 Purpose

G01EAF returns a one or two tail probability for the standard Normal distribution, via the routine name.

2 Specification

```

real FUNCTION G01EAF(TAIL, X, IFAIL)
  INTEGER          IFAIL
  real           X
  CHARACTER*1     TAIL

```

3 Description

The lower tail probability for the standard Normal distribution, $P(X \leq x)$ is defined by:

$$P(X \leq x) = \int_{-\infty}^x Z(X) dX$$

where

$$Z(X) = \frac{1}{\sqrt{2\pi}} e^{-X^2/2}, \quad -\infty < X < \infty.$$

The relationship

$$P(X \leq x) = \frac{1}{2} \operatorname{erfc} \left(\frac{-x}{\sqrt{2}} \right)$$

is used, where erfc is the complementary error function, and is computed using S15ADF. For the upper tail probability the relationship $P(X \geq x) = P(X \leq -x)$ is used and for the two tail significance level probability twice the probability obtained from the absolute value of x is returned.

When the two tail confidence probability is required the relationship

$$P(X \leq |x|) - P(X \leq -|x|) = \operatorname{erf} \left(\frac{|x|}{\sqrt{2}} \right),$$

is used, where erf is the error function, and is computed using S15AEF.

4 References

- [1] Hastings N A J and Peacock J B (1975) *Statistical Distributions* Butterworths
- [2] Abramowitz M and Stegun I A (1972) *Handbook of Mathematical Functions* Dover Publications (3rd Edition)

5 Parameters

1: TAIL — CHARACTER*1 *Input*

On entry: indicates which tail the returned probability should represent.

If TAIL = 'L', the lower tail probability is returned, i.e., $P(X \leq x)$.

If TAIL = 'U', the upper tail probability is returned, i.e., $P(X \geq x)$.

If TAIL = 'S', the two tail (significance level) probability is returned, i.e., $P(X \geq |x|) + P(X \leq -|x|)$.

If TAIL = 'C', the two tail (confidence interval) probability is returned, i.e., $P(X \leq |x|) - P(X \leq -|x|)$.

Constraint: TAIL = 'L', 'U', 'S' or 'C'.

2: X — *real* *Input*

On entry: the value of the standard Normal variate, x .

3: IFAIL — INTEGER *Input/Output*

On entry: IFAIL must be set to 0, -1 or 1. For users not familiar with this parameter (described in Chapter P01) the recommended value is 0.

On exit: IFAIL = 0 unless the routine detects an error (see Section 6).

6 Error Indicators and Warnings

If on entry IFAIL = 0 or -1, explanatory error messages are output on the current error message unit (as defined by X04AAF).

Errors detected by the routine: If IFAIL \neq 0, then G01EAF returns 0.0.

IFAIL = 1

On entry, TAIL \neq 'L', 'U', 'S' or 'C'.

7 Accuracy

Accuracy is limited by *machine precision*. For detailed error analysis see S15ADF and S15AEF.

8 Further Comments

None.

9 Example

Four values of TAIL and X are input and the probabilities calculated and printed.

9.1 Program Text

Note. The listing of the example program presented below uses bold italicised terms to denote precision-dependent details. Please read the Users' Note for your implementation to check the interpretation of these terms. As explained in the Essential Introduction to this manual, the results produced may not be identical for all implementations.

```
*      G01EAF Example Program Text
*      Mark 15 Release. NAG Copyright 1991.
*      .. Parameters ..
      INTEGER          NIN, NOUT
      PARAMETER       (NIN=5,NOUT=6)
*      .. Local Scalars ..
      real            PROB, X
      INTEGER         I, IFAIL
      CHARACTER       TAIL
*      .. External Functions ..
      real            G01EAF
      EXTERNAL        G01EAF
*      .. Executable Statements ..
      WRITE (NOUT,*) 'G01EAF Example Program Results'
*      Skip heading in data file
```

```

      READ (NIN,*)
      WRITE (NOUT,*)
      WRITE (NOUT,*) ' Tail      X      Probability '
      WRITE (NOUT,*)
      DO 20 I = 1, 4
        READ (NIN,*) TAIL, X
        IFAIL = 0
*
        PROB = G01EAF(TAIL,X,IFAIL)
*
        WRITE (NOUT,99999) TAIL, X, PROB
      20 CONTINUE
      STOP
*
99999 FORMAT (3X,A1,4X,F4.2,7X,F6.4)
      END

```

9.2 Program Data

G01EAF Example Program Data

```

'L'  1.96
'U'  1.96
'C'  1.96
'S'  1.96

```

9.3 Program Results

G01EAF Example Program Results

Tail	X	Probability
L	1.96	0.9750
U	1.96	0.0250
C	1.96	0.9500
S	1.96	0.0500
